

Laboratorio di Elettronica

Esercizi della Lezione 4:

Simulazione di circuiti con amplificatori operazionali

Valentino Liberali, Alberto Stabile



UNIVERSITÀ DEGLI STUDI DI MILANO
Dipartimento di Fisica "Aldo Pontremoli"


E-mail: valentino.liberali@unimi.it, alberto.stabile@unimi.it

Milano, 28-29 aprile 2022

- 1 Simulazione con SPICE dell'amplificatore operazionale ideale
- 2 Analisi nodale modificata per circuiti con generatori controllati
- 3 Modello in SPICE dell'amplificatore operazionale reale
- 4 Simulazione con SPICE dell'amplificatore operazionale reale
- 5 Circuito derivatore
- 6 Circuito integratore
- 7 Esercizi

Simulazione con SPICE dell'amplificatore operazionale – 1

L'amplificatore operazionale è un **generatore di tensione controllato in tensione**, che amplifica la differenza di tensione tra i due segnali V^+ e V^- .




Può essere descritto in SPICE come un amplificatore di tensione (**E**):

Exxxxxxxx N+ N- NC+ NC- GAIN

N+ e N- sono i nodi di uscita (N- è 0 se l'amplificatore operazionale ha un solo terminale di uscita); NC+ e NC- sono i nodi di ingresso; GAIN è il guadagno (che deve essere finito).

Simulazione con SPICE dell'amplificatore invertente – 1



* AMPLIFICATORE INVERTENTE (IDEALE)

```
VIN 1 0 SIN(0 10 100K 0 0)
```

```
R1 1 2 1K
```

```
R2 2 3 10K
```

```
EA 3 0 0 2 2E5
```

* EA è un amplificatore (quasi) ideale, con guadagno 200000

```
.OP
```


```
.TRAN 10N 50U
```

```
.PLOT TRAN V(1) V(3)
```

```
.END
```

Analisi nodale modificata con generatori controllati (1/2)


Amplificatore invertente realizzato con amplificatore operazionale e resistenze:



$$\left\{ \begin{array}{l} \textcircled{1} : -I_{V_X} - \frac{1}{R_1}(V_1 - V_2) = 0 \\ \textcircled{2} : \frac{1}{R_1}(V_1 - V_2) - \frac{1}{R_2}(V_2 - V_4) = 0 \\ \textcircled{3} : \frac{1}{R_3}(V_3 - 0) = 0 \\ \textcircled{E} : V_4 - 0 = E(V_3 - V_2) \\ \textcircled{V_X} : V_1 - 0 = V_X \end{array} \right.$$

L'equazione \textcircled{E} descrive l'amplificatore operazionale con guadagno finito E ; può essere riscritta in forma implicita come: $V_2 - V_3 + \frac{1}{E}V_4 = 0$

Analisi nodale modificata con generatori controllati (2/2)



$$\begin{bmatrix}
 \frac{1}{R_1} & -\frac{1}{R_1} & 0 & \vdots & 0 & \vdots & 1 \\
 -\frac{1}{R_1} & \left(\frac{1}{R_1} + \frac{1}{R_2}\right) & 0 & \vdots & -\frac{1}{R_2} & \vdots & 0 \\
 0 & 0 & \frac{1}{R_3} & \vdots & 0 & \vdots & 0 \\
 \vdots & \vdots & \vdots & \ddots & \vdots & \ddots & \vdots \\
 0 & 1 & -1 & \vdots & \frac{1}{E} & \vdots & 0 \\
 \vdots & \vdots & \vdots & \vdots & \vdots & \ddots & \vdots \\
 1 & 0 & 0 & \vdots & 0 & \vdots & 0
 \end{bmatrix} \cdot \begin{bmatrix} V_1 \\ V_2 \\ V_3 \\ \vdots \\ V_4 \\ \vdots \\ I_{V_X} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ \vdots \\ 0 \\ \vdots \\ V_X \end{bmatrix}$$

È sempre un'equazione matriciale nella forma $\mathbf{G} \cdot \mathbf{V} = \mathbf{I}$, ma \mathbf{G} non è simmetrica!

Sottocircuiti in SPICE – 1

In SPICE si può definire un “sottocircuito” (*subcircuit*), che poi verrà utilizzato una o più volte, come se fosse un singolo componente:

```
.SUBCKT NAME NI1 NI2 ...NIN
```

```
* netlist del sottocircuito
```

```
...
```

```
.ENDS
```

dove NAME è il nome dato al sottocircuito, e NI1 NI2 ...NIN è la lista dei terminali (i nodi che sono collegati verso l'esterno).

Esempio:

Useremo un modello di amplificatore operazionale reale definito come sottocircuito in questo modo:

```
* Model for uA741 Op Amp
```

```
.subckt uA741 1 2 3 4 5
```

```
c1 11 12 8.661E-12
```

```
c2 6 7 30.00E-12
```

```
dc 5 53 dy
```

```
...
```

```
.ends
```

Sottocircuiti in SPICE – 2

In SPICE il sottocircuito può essere utilizzato come componente solo dopo averlo definito.

Un sottocircuito è indicato con **X**:

XXXXXXXX N1 N2 ...NN NAME

I numeri dei nodi N1 N2 ...NN devono essere nello ordine dei terminali dichiarati nella riga .SUBCKT; NAME è lo stesso nome dato al sottocircuito nella riga .SUBCKT.

Esempio:

XOPAMP1 0 2 10 11 3 UA741

XOPAMP1 è il nome di questo componente;

0 e 2 sono i nodi di ingresso + e -;

10 e 11 sono le alimentazioni positiva e negativa;

3 è il nodo di uscita;

UA741 è il nome del sottocircuito (che deve essere stato definito in precedenza).

Modello dell'amplificatore operazionale reale – 1

```
* Model for uA741 Op Amp
* connections:
* 1 - non-inverting input
* 2 - inverting input
* 3 - positive power supply
* 4 - negative power supply
* 5 - output
*
.subckt uA741 1 2 3 4 5
*
c1 11 12 8.661E-12
c2 6 7 30.00E-12
dc 5 53 dy
de 54 5 dy
dlp 90 91 dx
dln 92 90 dx
dp 4 3 dx
```

(continua nella pagina successiva)

Modello dell'amplificatore operazionale reale – 2

(segue dalla pagina precedente)

```
egnd 99 0 poly(2),(3,0),(4,0) 0 .5 .5
fb 7 99 poly(5) vb vc ve vlp vln 0 10.61E6 -1E3 1E3 10E6 -10E6
ga 6 0 11 12 188.5E-6
gcm 0 6 10 99 5.961E-9
iee 10 4 dc 15.16E-6
hlim 90 0 vlim 1K
q1 11 2 13 qx
q2 12 1 14 qx
r2 6 9 100.0E3
rc1 3 11 5.305E3
rc2 3 12 5.305E3
re1 13 10 1.836E3
re2 14 10 1.836E3
ree 10 99 13.19E6
ro1 8 5 50
ro2 7 99 100
rp 3 4 18.16E3
```

(continua nella pagina successiva)

Modello dell'amplificatore operazionale reale – 3

(segue dalla pagina precedente)

```
vb 9 0 dc 0
vc 3 53 dc 1
ve 54 4 dc 1
vlim 7 8 dc 0
vlp 91 0 dc 40
vln 0 92 dc 40
.model dx D(Is=800.0E-18 Rs=1)
.model dy D(Is=800.00E-18 Rs=1m Cjo=10p)
.model qx NPN(Is=800.0E-18 Bf=93.75)
.ends
```

Il modello fa uso di generatori indipendenti di tensione (**V**) e di corrente (**I**), di generatori controllati (**E**, **F**, **G**, **H**), di resistenze (**R**) e capacità (**C**), di diodi (**D**) e di transistori bipolari a giunzione (**Q**).


Nel seguito, supponiamo che questo modello dell'amplificatore operazionale reale 741 sia stato scritto nel file: UA741.SPI

Modello dell'amplificatore operazionale reale – 4

Con questo modello per la simulazione dell'amplificatore operazionale 741, possiamo vedere gli effetti dovuti a:

- **tensione massima e minima in uscita:** la tensione V_{out} non può essere più alta dell'alimentazione positiva, o più bassa dell'alimentazione negativa;
- **guadagno finito:** il guadagno di tensione ha un valore finito:
- **banda finita:** il guadagno di tensione diminuisce all'aumentare della frequenza;
- **slew rate:** il valore assoluto della derivata della tensione di uscita rispetto al tempo $\frac{dV_{out}}{dt}$ non può superare un valore massimo (che è dovuto alla massima corrente in uscita).

Simulazione con SPICE dell'amplificatore invertente – 2



* AMPLIFICATORE INVERTENTE (REALE)

```
.INCLUDE UA741.SPI
```

```
VSP 10 0 10V
```

```
VSN 11 0 -10V
```

```
V1 1 0 SIN(0 1.5 1K 0 0)
```

```
R1 1 2 1K
```

```
R2 2 3 10K
```

```
XOA 0 2 10 11 3 UA741
```


```
.OP
```

```
.TRAN 1U 5M
```

```
.PLOT TRAN V(1) V(3)
```

```
.END
```

Circuito derivatore (1/2)



Per questo circuito, osserviamo anzitutto vale il principio della terra virtuale. Prendendo le correnti con il verso da sinistra verso destra, la KCL applicata all'ingresso (-) si può scrivere come $i_C(t) = i_R(t)$, da cui ricaviamo:


$$C \frac{d(v_{in}(t) - 0)}{dt} = \frac{0 - v_{out}(t)}{R}$$

e risolvendo rispetto a $v_{out}(t)$ si ottiene:

$$v_{out}(t) = -RC \frac{dv_{in}(t)}{dt} = -\tau \frac{dv_{in}(t)}{dt}$$

L'uscita è proporzionale alla *derivata* dell'ingresso.

Circuito derivatore (2/2)



Se $v_{in}(t) = V_A \sin 2\pi f_0 t$, allora:

$$v_{out}(t) = -2\pi f_0 R C V_A \cos 2\pi f_0 t =$$


$$v_{out}(t) = 2\pi f_0 R C V_A \sin \left(2\pi f_0 t + \frac{3\pi}{2} \right)$$

L'ampiezza della tensione è moltiplicata per $2\pi f_0 R C = 2\pi f_0 \tau$.

Nel dominio della frequenza, il guadagno aumenta in modo direttamente proporzionale alla frequenza.

Il derivatore ha una risposta in frequenza il cui diagramma di Bode del modulo ha una pendenza costante di $+20 \text{ dB/decade}$.

Circuito integratore (1/2)




Il circuito integratore si risolve in modo analogo al precedente.

Dalla KCL al nodo di terra virtuale si ha $\frac{v_{in}(t)}{R} = -C \frac{dv_{out}(t)}{dt}$, e risolvendo rispetto a $v_{out}(t)$ si ottiene:

$$v_{out}(t) = -\frac{1}{RC} \int_0^t v_{in}(t) dt + v(0)$$

L'uscita è proporzionale all'*integrale* dell'ingresso.

Circuito integratore (2/2)



Se $v_{in}(t) = V_A \sin 2\pi f_0 t$, allora:

$$v_{out}(t) = \frac{V_A}{2\pi f_0 RC} \cos 2\pi f_0 t + v(0) = \frac{V_A}{2\pi f_0 RC} \sin \left(2\pi f_0 t + \frac{\pi}{2} \right) + v(0)$$

L'ampiezza della tensione è divisa per $2\pi f_0 RC = 2\pi f_0 \tau$.

Quindi l'integratore ha una risposta in frequenza che decresce: il diagramma di Bode del modulo ha una pendenza costante di -20 dB/decade.

Esercizi proposti

- ❶ Simulare con Ngspice l'amplificatore invertente e un amplificatore non invertente, verificandone il funzionamento nel dominio del tempo e nel dominio della frequenza.
 - ▶ Verificare (su un circuito a scelta) che con la simulazione nel tempo si possono osservare gli effetti dovuti alla limitazione della tensione in uscita e allo slew rate.
 - ▶ Verificare (su un circuito a scelta) che con la simulazione in frequenza si possono osservare gli effetti dovuti alla banda finita.
- ❷ Simulare con Ngspice un derivatore e un integratore:
 - ▶ Verificare il comportamento nel tempo, al variare delle frequenza;
 - ▶ Ricavare i diagrammi di Bode mediante simulazione con Ngspice;
 - ▶ (opzionale) Ripetere l'esercizio per diversi valori di resistenza, capacità e modello di amplificatore.
 - ▶ Per quali circuiti si notano gli effetti dovuti al guadagno finito e alla banda finita dell'amplificatore operazionale reale?